

Lenguajes de Programación y Procesadores de Lenguajes 2014/15

INTRODUCCIÓN A BISON

Escuela Técnica Superior de Ingeniería Informática
Universitat Politècnica de València

V.14.0

Índice

1. Presentación	1
2. Descripción de un analizador sintáctico mediante Bison	2
2.1. Sección de declaraciones	2
2.1.1. Declaraciones en C	2
2.1.2. Declaraciones de Bison	2
2.2. Sección de Reglas Gramaticales	2
2.3. Código C adicional	3
3. Integración Flex-Bison.	4
3.1. Generación de Analizadores Léxico-Sintácticos	4
3.2. Adaptación de un analizador léxico para integrarse con Bison	5

1. Presentación

Bison es un generador automático de analizadores sintácticos que también permite asociar acciones semánticas a las reglas gramaticales. Tal y como muestra la Figura 1 Bison recibe como entrada un fichero de texto (que por convención lleva la extensión `.y`) con las reglas de una gramática independiente del contexto y genera código C que implementa el analizador sintáctico LALR(1) correspondiente a la especificación proporcionada.

Un analizador sintáctico necesita que un analizador léxico le proporcione los símbolos léxicos. Este analizador léxico puede implementarse por el programador o usarse uno generado por la herramienta Flex tal y como se verá en la sección 3.




Figura 1: Uso de Bison

2. Descripción de un analizador sintáctico mediante Bison

Un fichero fuente de Bison se divide, al igual que un fichero fuente de Flex, en tres secciones separadas por una línea donde aparecen únicamente los caracteres `%%` :

1. **Declaraciones:** En esta sección aparece una subsección destinada a realizar declaraciones en código C delimitada mediante los caracteres `%{` y `%}` y otra destinada a declaraciones propias de Bison.
2. **Reglas Gramaticales:** En esta sección se definen las reglas gramaticales que constituyen la especificación sintáctica.
3. **Código C adicional:** En esta sección opcional se definen funciones C que se incluirán en el fichero C generado.

A continuación se presentan cada una de estas secciones en detalle.

2.1. Sección de declaraciones

Esta sección se encuentra dividida en dos subsecciones: declaraciones escritas en el lenguaje C y declaraciones propias de Bison.

2.1.1. Declaraciones en C

La sección de declaraciones en C contiene definiciones de macros y declaraciones de funciones y variables que se utilizan en las acciones asociadas a las reglas de la gramática.

A continuación vemos un ejemplo en el que se utiliza la directiva `#include` para incluir el archivo de cabecera `stdio.h` además de la declaración de la variable externa `int yylineno`. La variable `int yylineno` contiene los números de las líneas que se van procesando y está definida en el analizador léxico generado por Flex. Podemos utilizarla gracias a la integración de ambas herramientas que se verá en la Sección 3.

```
%{  
#include <stdio.h>  
extern int yylineno;  
%}
```

2.1.2. Declaraciones de Bison

La sección de declaraciones de Bison contiene la definición de los símbolos terminales (tokens o símbolos léxicos) y, opcionalmente, los no-terminales de la gramática. Normalmente todos los tokens declarados en esta sección deberán ser identificados por el analizador léxico utilizado. A modo de ejemplo, se muestra la declaración de 4 tokens:

```
%token INT_ FLOAT_  
%token ID_ PTOCOMA_
```

Adicionalmente también puede aparecer una declaración `%error-verbose` para activar la verbosidad en la detección de errores (se muestran mensajes de error más detallados durante el análisis sintáctico).

2.2. Sección de Reglas Gramaticales

La sección de las reglas gramaticales contiene las reglas de la gramática que define el lenguaje que se desea procesar. Debe haber siempre al menos una regla gramatical.

Una regla gramatical de Bison tiene la siguiente forma general:

```
no-terminal: lado_derecho {acciones_semánticas};
```

donde las acciones semánticas (opcionales) estarán formadas por código C que se ejecutará cuando se aplique la reducción a la que están asociadas¹.

Para indicar que un no-terminal produce la cadena vacía, basta con dejar el lado derecho de una de sus producciones vacío. Para separar las distintas producciones de un mismo no-terminal se usa la barra vertical |. El símbolo punto y coma (;) aparece al final de la última producción de cada no-terminal.

Por ejemplo, si una gramática tiene las siguientes reglas para el no-terminal S:

$$S \rightarrow \varepsilon \mid abA \mid E$$

su implementación en Bison quedaría:

```
S: | a b A | E;
```

que significa que S produce “cadena vacía”, “abA” y “E”.

En Bison se asume por defecto que el símbolo inicial de la gramática es el primer no-terminal que se encuentra en la sección de especificación de la gramática. Si se desea especificar explícitamente cuál es el símbolo inicial se debe usar la declaración %start:

```
%start programa
```

A continuación se muestran las reglas que definen la sintaxis de declaraciones de variables simples en un lenguaje estructurado de alto nivel (similar a como se haría en el lenguaje C).

```
secuenciaDeclaraciones: declaracion
                        | secuenciaDeclaraciones declaracion ;
declaracion: tipo ID_ PTOCOMA_ ;
tipo: INT_ | FLOAT_;
```

2.3. Código C adicional

En esta sección opcional se incluyen funciones escritas por el usuario en el lenguaje C que se incluirán en el fichero generado por Bison. Estas funciones podrán invocarse desde las acciones de las reglas.

Cada vez que el analizador sintáctico generado por Bison detecta un error llamará a la función yyerror. La implementación de esta función debe realizarla el usuario/programador de Bison y puede incluirse en esta sección o en algún fichero externo. Una sencilla implementación de esta función podría ser la siguiente:

```
yyerror (char *s) {
    printf ("Linea%d:%s", yylineno, s);
}
```

A modo de ejemplo se muestra a continuación el código completo de la especificación Bison para un sencillo lenguaje que reconoce la declaración de variables sencillas en un lenguaje de programación de alto nivel similar a C.

```
%{
#include <stdio.h>
extern int yylineno;
%}
```

```
%error-verbose
```

```
%token ID_ PTOCOMA_
```

¹La definición y uso de estas acciones no forma parte de este manual introductorio a Bison.

```

%token INT_ FLOAT_

%%

secuenciaDeclaraciones: declaracion
                        | secuenciaDeclaraciones declaracion
                        ;
declaracion: tipo ID_ PTOCOMA_
            ;
tipo: INT_
     | FLOAT_
     ;

%%

/* Llamada por yyparse ante un error */
yyerror (char *s)
{
    printf ("Linea %d: %s\n", yylineno, s);
}

```

3. Integración Flex-Bison.

3.1. Generación de Analizadores Léxico-Sintácticos

El analizador sintáctico generado por Bison necesita interactuar con un analizador léxico para su correcto funcionamiento. El analizador léxico debe encargarse de analizar los lexemas de la entrada con el fin de identificar aquellos que corresponden con los tokens definidos en la sección de declaraciones de Bison y que se utilizan como terminales en las reglas gramaticales.

El proceso de análisis de un analizador sintáctico generado por Bison se inicia llamando a la función `yyparse()`. Esta función se encarga de llamar a la función `yylex()` que debe contener la implementación del analizador léxico. El analizador léxico (`yylex()`) puede implementarse en la sección de código adicional de Bison o en un fichero externo. Esta última opción constituye el punto clave de la integración Flex-Bison, ya que Flex implementa un analizador léxico mediante una función `yylex()`.

Al mismo tiempo, si el analizador léxico generado por Flex debe identificar los tokens definidos en Bison, se debe utilizar algún mecanismo que permita compartir estas definiciones. Dado que tanto el analizador léxico como el sintáctico están implementados en C, este mecanismo será un fichero de cabecera.

A modo de resumen en la Figura 2 se muestra cómo se realiza la integración Flex-Bison: Por un lado Bison se encarga de generar el analizador sintáctico que contiene la función `yyparse()`. También se encarga de generar un fichero de cabecera que le permitirá a Flex conocer los tokens definidos. Por otro lado, Flex se encarga de generar el analizador léxico que incluye la función `yylex()`, en cuya implementación se incluirá el fichero de cabecera.

Pasos para la integración Flex-Bison:

1. En primer lugar se debe ejecutar la herramienta Bison pasándole como argumento el nombre del fichero donde se encuentra la especificación sintáctica. Por convención, estos ficheros llevan la extensión `.y` (por ejemplo, `asin.y`). Además, para indicarle a Bison que debe generar el archivo de cabecera que permitirá compartir las declaraciones de tokens con Flex, se debe utilizar la opción `-d`. Si la especificación es correcta Bison genera dos ficheros:
 - Por un lado el fichero C que constituye la implementación del analizador sintáctico. El nombre de este fichero se construye reemplazando la extensión `.y` del nombre del fichero fuente Bison por la extensión `.tab.c`.




Figura 2: Integración de Bison con Flex

- Bison también genera el fichero de cabecera necesario para integrarse con Flex. El nombre de este fichero se construye reemplazando la extensión `.y` del nombre del fichero fuente Bison por la extensión `.tab.h`.

De este modo, a partir de un fichero Bison llamado `asin.y`, se obtendrían los ficheros `asin.tab.c` y `asin.tab.h`.

```
> bison -d asin.y
```

En caso de querer cambiar el nombre de los ficheros generados se puede usar la opción `-o` seguida del nombre del fichero C. En la llamada a Bison del siguiente ejemplo se generarán los ficheros `asin.c` y `asin.h`:

```
> bison -oasin.c -d asin.y
```

2. Se ejecuta la herramienta Flex para generar la implementación del analizador léxico². Hay que tener en cuenta que se debe incluir el fichero `.h` generado por Bison en la especificación léxica de Flex (fichero `.l`).

```
> flex -oalex.c alex.l
```

3. Los ficheros C generados por Bison y Flex se compilan y enlazan con la librería `lfl` para obtener el ejecutable de un analizador léxico-sintáctico.

```
> gcc -omianalizador alex.c asin.c -lfl
```

4. Se puede ejecutar el analizador léxico-sintáctico para que analice la entrada estándar (sin argumentos) o el texto de un fichero (pasado por argumento) si se ha preparado el analizador léxico para leer y usar este argumento de entrada.

```
> mianalizador ejemplo.c
```

3.2. Adaptación de un analizador léxico para integrarse con Bison

Al integrar Bison con Flex, el analizador léxico generado por este último debe encargarse de analizar la cadena de entrada, identificar los tokens definidos en Bison y comunicar al analizador sintáctico la detección de dichos tokens. También debe encargarse de iniciar el análisis léxico-sintáctico llamando a la función `yyparse`. Si ya se dispone de un analizador léxico y se desea integrarlo con el analizador sintáctico generado por Bison, deben realizarse las siguientes tareas:

²Para más información de la compilación con Flex consultar la documentación de esta herramienta.

1. Incluir el fichero de cabecera generado por Bison en la sección de declaraciones C de Flex.
2. Las acciones de las reglas léxicas deben informar a Bison de los tokens identificados mediante la sentencia `return` seguida del token reconocido.
3. La función `main` debe llamar al analizador sintáctico: `yyparse()`.

A modo de ejemplo se muestra a continuación la especificación Flex necesaria para generar un analizador léxico que pueda integrarse con el analizador sintáctico del ejemplo de la declaración de variables mostrado anteriormente:

```
%{
#include <stdio.h>
#include "asin.h"
extern FILE *yyin;
%}

%option yylineno

ident [a-zA-Z][a-zA-Z0-9]*
delimitador [ \t\n]+

%%

{delimitador} { }
{ident}      {return ID_;}
"int"        {return INT_;}
"float"      {return FLOAT_;}
";"          {return PTOCOMA_;}
.            {yyerror("Caracter Desconocido");}

%%

int main (int argc, char **argv)
{
    if ((yyin = fopen (argv[1], "r")) == NULL)
        fprintf (stderr, "Fichero no valido \\\%s", argv[1]);
    yyparse();
}
```